Section 2

In this notebook we provide an example of computing the scattering amplitude of $M_{0,5}$ using the Cachazo-He-Yuan formula (https://arxiv.org/abs/1306.6575)
The following is the log-likelihood function on $M_{0,5}$, also called the scattering potential in physics literature.

In[9]:=
```
logLikelihood = s1 Log[x] + s2 Log[y] + s3 Log[1 - x] + s4 Log[1 - y] + s5 Log[y - x];
```

We symbolically solve for the critical points of the log-likelihood function. They are the points where the partial derivatives vanish.

In[10]:=
```
{D[logLikelihood, x] == 0, D[logLikelihood, y] == 0}
```

Out[ ]=
$$\left\{\frac{s_1}{x} - \frac{s_3}{1-x} - \frac{s_5}{-x+y} == 0,\ \frac{s_2}{y} - \frac{s_4}{1-y} + \frac{s_5}{-x+y} == 0\right\}$$

In[27]:=
```
critPts = FullSimplify[
  Solve[{D[logLikelihood, x] == 0, D[logLikelihood, y] == 0}, {x, y}]]
```

Out[ ]=
$$\Big\{\Big\{x \to \frac{1}{2\,(s_1+s_3+s_5)\,(s_1+s_2+s_3+s_4+s_5)}\Big(2 s_1^2 + s_2\,(s_3+s_5) + s_5\,(s_3+s_4+s_5) + s_1\,(2 s_2 + 2 s_3 + s_4 + 3 s_5) - \sqrt{\Big(s_2^2\,(s_3+s_5)^2 + (s_1\,(s_4+s_5) + s_5\,(s_3+s_4+s_5))^2 + 2 s_2\,\big(s_5\,(s_3+s_5)\,(s_3+s_4+s_5) + s_1\,(-s_3 s_4 + (s_3+s_4)\,s_5 + s_5^2)\big)\Big)}\Big),$$
$$y \to \frac{1}{2\,(s_2+s_4+s_5)\,(s_1+s_2+s_3+s_4+s_5)}\Big(2 s_2^2 + s_1\,(2 s_2+s_4+s_5) + s_5\,(s_3+s_4+s_5) + s_2\,(s_3 + 2 s_4 + 3 s_5) + \sqrt{\Big(s_2^2\,(s_3+s_5)^2 + (s_1\,(s_4+s_5) + s_5\,(s_3+s_4+s_5))^2 + 2 s_2\,\big(s_5\,(s_3+s_5)\,(s_3+s_4+s_5) + s_1\,(-s_3 s_4 + (s_3+s_4)\,s_5 + s_5^2)\big)\Big)}\Big)\Big\},$$
$$\Big\{x \to \frac{1}{2\,(s_1+s_3+s_5)\,(s_1+s_2+s_3+s_4+s_5)}\Big(2 s_1^2 + s_2\,(s_3+s_5) + s_5\,(s_3+s_4+s_5) + s_1\,(2 s_2 + 2 s_3 + s_4 + 3 s_5) + \sqrt{\Big(s_2^2\,(s_3+s_5)^2 + (s_1\,(s_4+s_5) + s_5\,(s_3+s_4+s_5))^2 + 2 s_2\,\big(s_5\,(s_3+s_5)\,(s_3+s_4+s_5) + s_1\,(-s_3 s_4 + (s_3+s_4)\,s_5 + s_5^2)\big)\Big)}\Big),$$
$$y \to \frac{1}{2\,(s_2+s_4+s_5)\,(s_1+s_2+s_3+s_4+s_5)}\Big(2 s_2^2 + s_1\,(2 s_2+s_4+s_5) + s_5\,(s_3+s_4+s_5) + s_2\,(s_3 + 2 s_4 + 3 s_5) - \sqrt{\Big(s_2^2\,(s_3+s_5)^2 + (s_1\,(s_4+s_5) + s_5\,(s_3+s_4+s_5))^2 + 2 s_2\,\big(s_5\,(s_3+s_5)\,(s_3+s_4+s_5) + s_1\,(-s_3 s_4 + (s_3+s_4)\,s_5 + s_5^2)\big)\Big)}\Big)\Big\}\Big\}$$

To compute the scattering amplitude using the CHY formula, we need two ingredients. The first is the Hessian determinant of the log-likelihood function. The second is called the Parke-Taylor factor PT = $\frac{1}{p_{12}\,p_{23}\,p_{34}\,p_{45}\,p_{51}}$.

We first compute the Hessian.

In[12]:=
```
H = Table[D[D[logLikelihood, w], z], {w, {x, y}}, {z, {x, y}}];
MatrixForm[H]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{s_1}{x^2}-\frac{s_3}{(1-x)^2}-\frac{s_5}{(-x+y)^2} & \frac{s_5}{(-x+y)^2} \\ \frac{s_5}{(-x+y)^2} & -\frac{s_2}{y^2}-\frac{s_4}{(1-y)^2}-\frac{s_5}{(-x+y)^2} \end{pmatrix}$$

We use the following parametrization for $M_{0,5}$.

In[14]:=
```
M = Transpose[{{1, 1, 1, 1, 0}, {0, x, y, 1, 1}}];
Minors[M, 2]
```

Out[∘]= {{x}, {y}, {1}, {1}, {-x + y}, {1 - x}, {1}, {1 - y}, {1}, {1}}

The Parke-Taylor factor is $\mathrm{PT} = \frac{1}{p_{12}\,p_{23}\,p_{34}\,p_{45}\,p_{51}}$.

In[∘]:=
```
Times @@ Map[Det[M[[#]]]^-1 &, {{1, 2}, {2, 3}, {3, 4}, {4, 5}, {5, 1}}]
```

Out[∘]= $-\frac{1}{x\,(1-y)\,(-x+y)}$

In[17]:= $\mathrm{PT} = \frac{1}{x\,(-x+y)\,(1-y)}$;

The scattering amplitude is computed as the sum over the critical points of the log-likelihood function of the integrand we described before

$\sum_{\mathrm{critPts}}\left(\frac{1}{\mathrm{Det}[H]}\ \mathrm{PT}^2\right)$

It can be simplified to a rational expression in terms of s_i, namely expression (10) in the paper. However, this computation is very slow symbolically, so we give a numerical confirmation that the scattering amplitude is indeed equal to expression (10).
We plug in numerical values for s_i.

In[18]:=
```
svar = {s1, s2, s3, s4, s5};
svalue = RandomInteger[{1, 50}, {5}]
ncritPts = critPts /. Thread[svar → svalue]
```

Out[∘]= {33, 38, 23, 24, 40}

Out[∘]=

$$\left\{\left\{x\to\frac{16830-6\sqrt{1694649}}{30336},\ y\to\frac{18246+6\sqrt{1694649}}{32232}\right\},\ \left\{x\to\frac{16830+6\sqrt{1694649}}{30336},\ y\to\frac{18246-6\sqrt{1694649}}{32232}\right\}\right\}$$

In[21]:=
```
nH = H /. Thread[svar → svalue];
amplitude =
  Simplify[(1/Det[nH] PT^2 /. ncritPts[[1]]) + (1/Det[nH] PT^2 /. ncritPts[[2]])]
```

$$\text{Out[}\circ\text{]=}\quad \frac{5369}{2\,124\,540}$$

We evaluate expression (10):

In[23]:=
```
expression10 =
  1/(s1 s4) + 1/(s4 (s3 + s4 + s5)) + 1/((s3 + s4 + s5) s5) + 1/(s5 (s1 + s2 + s5)) + 1/((s1 + s2 + s5) s1);
Simplify[expression10 /. Thread[svar → svalue]]
```

$$\text{expression10} = \frac{1}{s_1 s_4} + \frac{1}{s_4 (s_3 + s_4 + s_5)} + \frac{1}{(s_3 + s_4 + s_5) s_5} + \frac{1}{s_5 (s_1 + s_2 + s_5)} + \frac{1}{(s_1 + s_2 + s_5) s_1};$$

$$\text{Out[}\circ\text{]=}\quad \frac{5369}{2\,124\,540}$$

We see that the amplitudes computed in the two ways agree.